
Глава 5   Ой - ООП?
(Прим. перев.: Перевод у меня конечно ужасный, так что кто не поймет, я советую прочитать 3-ю главу первой книги Колонина «Основы программирования в CA-Visual Objects» )
В этой главе, мы возьмем перерыв в процессе практического создания приложений, чтобы изучить средства Объектно-Ориентируемого Программирования. Не волнуйтесь, если Вы не сразу все поймете при первом прочтении; Вы будете понимать это по мере того, как изучите еще несколько глав. Тогда может будет полезно возвратиться к этой главе, чтобы закрепить ваше понимание.

Объектно-Ориентируемое Программирование (ООП) появилось не за один день в результате озарения одного человека. ООП появился в 1960-ых годах в Европе, но вошел со всеми своими сложностями управления событиями в многозадачном интерфейсе Windows GUI. У ООП было бедное начало, потому что не было ни одного языка, который бы поддерживал его должным образом при наличии достаточных встроенных функциональных возможностей. В CA-VISUAL OBJECTS разработанное приложение это объект и ООП используется в поддержке GUI интерфейса, управлении ошибками, управлении базами данных и отчетов. То, насколько ООП используется в поддержке бизнес-логики приложения, полностью зависит от разработчика
Как концепция, ООП основывается на четырех принципах – 

· Абстракция. Способность создавать типы данных соответствующие к идеям будучи представленные как данные. 
· Инкапсуляция. Способность скрывать детали выполнения созданных типов данных. 
· Наследование. Способность создавать новые типы данных от существующих типов данных. 
· Полиморфизм. Способность использовать одно и то же имя для решения двух или более схожих, но технически разных задач
Эти четыре принципа должны быть включены в язык программирования, чтобы можно было назвать этот язык языком Объектно-Ориентированным программирования. CA-Visual Objects - один из очень немногих языков в  котором встречают все эти критерии.

Абстракция
Абстракция относится к процессу реферирования, попросту говоря, отделения обязательных свойств от реального объекта жизни и создания типа данных (объектный тип или класс объектов) с соответствующими свойствами.

Накладная - типичный реальный объект жизни. В этом случае будет не трудно  абстрагировать его свойства - 
· Номер Накладной, Дата 
· Имя Заказчика, адрес, номер 
· Линии Накладной  с названием изделия , количество, цена, процентная ставка налога, налог 
· Итого: Налог и Сумма 

Мы можем создать новый тип данных для всего вышеперечисленного, опредеть Класс и все объекты этого класса будут содержать определенные переменные для вышеупомянутых деталей. Обратите внимание, что Линии Накладной непосредственно осуществлены как объекты другого класса (InvoiceLine Класс) и Класс Накладной будет иметь массив свойств способным содержать так много объектов InvoiceLine сколько будет необходимо. Можно было собрать вместе все вышеописанные детали данных для одного счета и сохранить их в массиве или в структуре на не-ООП языках, но выполнение всего этого направлением Класс-Объект намного лучше, далее мы это сможем увидеть.

Писать программы не-ООП способом, также включает абстрагирование данных, то есть проходит вышеупомянутый процесс. Различие в том, что, в ООП, все это может быть собранно в один пакет называемый по другому объектом. Лучше всего, мы можем создать класс, который определит типы данных и имеет много экземпляров этого класса как индивидуальные объекты хранящиеся в памяти. Например, создав Класс Накладная выше, мы можем создать так много экземпляров этого класса, (как много накладных) сколько нам потребуется.

Пока, мы только рассмотрели, что значения расценены как часть нашего типа данных. Он также может связать процессы, также как числовой тип данных может связать процессы (добавить, вычесть, умножить, делить и т.д.). Для Класса Накладная, связанные процессы могли бы быть Печать, Флажок что оплачено, Выявить злостных дебиторов и т.д. Эти связанные процессы - подобно функциям за исключением того, что они привязаны к классу и называются Методами. Классы и структуры схожи, но структурам не могут принадлежать связанные методы. Остальная часть этой главы сконцентрируется на классах.

Наша Накладная может быть назначена переменной точно так же, как любой другой тип данных –

oInvoice: = InvoiceClass{}

В Главе 1 мы точно таким же способом создали объекты окна –

oShell: = MyShell{ SELF }

хотя SELF был дополнительным компонентом, мы подойдем к нему позже.

Создав наш собственный тип данных для накладной, мы можем повторно использовать этот тип данных в других прикладных программах. Очевидно, что это может быть очень полезно, но это также подразумевает, что мы должны сделать усилие, чтобы спроектировать тип данных для большего количества общей применимости. Но об этом чуть позже.

Класс определяется очень просто, как вот тут -

CLASS Train

EXPORT nPassengers AS DWORD

EXPORT nMaxPassengers AS DWORD
Здесь мы определили класс только с двумя свойствами. Обе - переменные, потому что каждый экземпляр класса будет иметь различную пару значений, переменные называются Переменные Экземпляров или IVars. Они были объявлены DWORD в этом случае, а слово EXPORT определяет их видимость, но об этом тоже по позже.
Мы можем обратиться к Ivars в объектах этого класса, указывая объектный идентификатор, сопровождаемый двоеточием (Посылающий Оператор), сопровожденный IVAR как в <oTrain:nPassengers>.

Синтаксис - 

<objects>:<IVar>

В следующем сегменте кода, мы во-первых определяем переменную <oTrain>, которая может назначить только экземпляр этого Класса Train. Мы это делаем на следующей линии, где экземпляр класса Train присвоен переменной <oTrain>. Заметьте фигурные скобки, правильный синтаксис для создания экземпляра класса, используемые после имени класса.

LOCAL oTrain AS Train

...

oTrain := Train{}

oTrain:nPassengers := 2000

oTrain:nMaxPassengers := 1500

На следующих двух линиях, мы присваиваем значения двум Переменным Экземпляра, связанные с объектом, <oTrain:nPassengers> и <oTrain:nMaxPassengers>. Факт, что теперь на борту больше пассажиров чем, там должно быть, значит  мы должны  сделать все чтобы предотвратить такой случай.

Мы только рассмотрели один поезд. Их может быть много и каждый должен быть идентифицирован. Каждый должен иметь различные имена переменных как  <oTrain2>, <oMyTrain>, <o10amToLondonTrain>, и т.д. Но каждый имеет тот же самый диапазон свойств и методов, поскольку они были все определены в одном Классе Train. Мы увидим позже, что хранение групп объектов в массивах - гораздо более эффективный путь идентификации их.

Мы можем написать функции, которые имеют определенные отношения с классом, в этом случае они называются методами класса, как в –

METHOD NewLeg(nGettingOn, nGettingOff) CLASS Train

// Changes the number of current passengers

LOCAL dwPass as DWORD

LOCAL lSuccess as LOGIC

dwPass := SELF:nPassengers + nGettingOn - nGettingOff

IF dwPass > SELF:nMaxPassengers

   TextBox{ , “Passenger Number Validation”, ;

            “Too many for Train by ” ;

            + NTrim(dwPass - SELF:nMaxPassengers)}:Show()

ELSEIF dwPass < 0

   TextBox{ , “Passenger Number Validation”, ;

           “Less than zero Passengers”}:Show()

ELSE

   SELF:nPassengers := dwPass // Updates the IVar

   lSuccess := TRUE

ENDIF

RETURN lSuccess
К вышеупомянутому методу обратились бы с синтаксисом –

oTrain:NewLeg(22, 21)

или, в синтаксических терминах –

<object>:<method([Argument List])>
почти так же как вызвана Ivars в –

<object>:<Ivar>
Во-первых, обратите внимание на использование зарезервированного слова SELF, который обращается к объекту, которому было послано сообщение. <SELF>, в этом случае, является Объектом Train, к которому этот метод применяется. <SELF:nPassengers>, поэтому, <nPassengers> это Ivar обращающийся к  текущему объекту класса Train.

<SELF> всегда обращается к объекту и только появляется в коде методов. <SELF:nPassengers> имеет очевидное подобие <oTrain:nPassengers>, замеченному ранее; если <oTrain> - объект текущего метода, тогда <SELF>, тот же самое что и <oTrain>.

Мы имеем дело с тем же самым поездом и Вы помните, что <oTrain:nMaxPassengers> было присвоено 1500, так что, если будет больше пассажиров, чем нужно, то этот метод возвращает FALSE.

Теперь давайте посмотрим на работу вышеупомянутого метода. Определение метода (первая линия) указывает его имя, параметры <nGettingOn> и <nGettingOff>,  позволяют внешней информации проходить в метод, и имя класса, которому он принадлежит. Затем мы определи две ЛОКАЛЬНЫЕ переменные <dwPass> и <lSuccess>, который будет использоваться в течение обработки метода и отвергнуты, когда метод закончится.

Следующая линия присваивает значение к <dwPass>. Это значение - текущее значение Ivar <SELF:nPassengers> с добавленным значением в параметре <nGettingOn>  и вычтенным значение в <nGettingOff>, составляют новое число пассажиров, сохраненное во временной (ЛОКАЛЬНОЙ) переменной <dwPass>.

Следующие несколько линий определяют, является ли значение <dwPass> приемлемым. Метод построил правила, чтобы проверить это значение перед заменой текущего значения Ivar < SELF:nPassengers > на новое значение. Это значение не может быть больше, чем максимально позволено и не может быть меньше нуля; если любая из этих ситуаций встречается, появляется сообщение и изменение отклоняется, поскольку новое значение не может отражать действительность. Если новое значение проходить обе проверки, Ivar изменяется и <lSuccess> устанавливается в TRUE. Наконец, метод ВОЗВРАЩАЕТ значение <lSuccess>, TRUE или FALSE.

Код программы, который вызывает метод должен 'знать', что делать в зависимости от TRUE или FALSE пришедший назад к нем от инструкции метода RETURN. (Мы не должны устанавливать альтернативное значение FALSE для < lSuccess >, поскольку это - значение присваивается по умолчанию для недавно созданной переменной типа ЛОГИКА.)

Мы уже видели, что присваиваемые значения к Ivars <nPassengers> и <nMaxPassengers> были неудовлетворительны. Это можно преодолеть, используя метод Init() для класса, который обеспечивает начальные значения для его Ivars. Все классы должны иметь Init() или экземпляр метода. Некоторые языки предпочитают называть его Методом Создателя. Его используют, чтобы создать объект некоторого 'вещества'. В этом случае, следующее должно быть достаточным -

METHOD Init(nMaxPass) CLASS Train

// Set a value for the maximum number of passengers allowed on

// the train ...

SELF:nMaxPassengers := nMaxPass

// This line is not strictly needed as it starts as 0 anyway ...

SELF:nPassengers := 0

С этим методом Init(), экземпляр объекта поезда требует различного синтаксиса от используемого прежде -

oTrain1:= Train{2000}

Заметьте, что предварительно мы приписали значение к объекту поезда при помощи  фигурных скобок. Здесь, мы включили значение между фигурными скобками, который передается как параметр для метода Init(), где он отображен как параметр <nMaxPass> и затем присвоен к Ivar <nMaxPassengers> класса.

Обратите внимание, что вышеупомянутый метод Init() не имеет никакого оператора RETURN. Если Вы не включаете ни одного Визуального Объекта, по умолчанию добавляется линия -

RETURN SELF

Мы можем захотеть возвратить определенное значение, как мы это делали в методе NewLeg() выше. Причина возврата <SELF> - то, что методы могут быть подключены в одну инструкцию, больше об этом чуть позже. Обратите внимание на особенность, что методы Init() должны всегда возвращать <SELF> (или полагаться на значение по умолчанию, чтобы заполнить его для Вас).

Каждый не должен использовать методы, чтобы обратиться или изменить свойства класса. Мы можем теперь установить фактическое число пассажиров, повторя вышеупомянутый код с одной дополнительной линией -

oTrain1:= Train{2000}

oTrain1:nPassengers := 234

но мы теперь видим, что у нее есть недостатки.

В предыдущих главах, мы приписали значение к объектам, для которых необходимые классы существовали в пределах CA-Visual Objects, например, ShellWindow и DB Server. Здесь мы создали новый класс, который не имел никакого происхождения в пределах Визуально-объектного языка; мы создали абстракцию.

Инкапсуляция
Внутренние работы типа данных, включая Абстрактный Тип данных (Определенный пользователем Тип данных), не должны быть известны программисту, использующих их. Они, возможно, были куплены как часть библиотеки и они, возможно, даже были написаны на другом языке. Это означает, что внутренние работы нового типа данных могут быть изменены его автором не влияя на его использование другими, пока его существующий интерфейс (появление от внешней стороны) остается тем же самым. Такое изменение могло только удалять ошибку или добавлять дополнительные свойства или процессы на интерфейсе. Пока существующие внешне видимые свойства и процессы остаются и все еще функционируют правильно, не имеет значения, что другие свойства и процессы, возможно, были добавлены для использования другими прикладными программами или измененной версией той же самой прикладной программой.

Абстрактный тип данных был определен как класс из которого разработчик может проиллюстрировать объекты как требуется. Разработчик должен знать об интерфейсе обеспеченным классом: то есть, его экспортируемые переменные экземпляров и методы.

Класс также, может иметь переменные экземпляры и методы, которые - не видимы в интерфейсе и поэтому не могут использоваться разработчиком при использовании класса. Одна причина для утаивания переменных экземпляров от разработчиков состоит в том, чтобы осуществить проверку правильности изменений в переменных экземпляров. Сделать это можно делать с помощью модификации значения, и только через метод (или специальный тип метода называемый методом ASSIGN - увидит ниже). Эта способность к инкапсулировать свойства (или скрывать их от прямой манипуляции) – основная особенность любого языка, требующая быть ООП языком. Если Класс Train имел данные HIDDEN Ivars вместо EXPORT, то линия -

oTrain1:nPassengers := ...

не работала бы, поскольку <nPassengers> будет вне области. Конечно, изменение числа пассажиров не должен быть недоступным; мы - только накладываем большее количество управления на использовании класса.

Мы обсудили методы обычным путем при помощи Абстракции. Однако, есть два типа метода, которые глубоко связаны с понятием инкапсуляции: методы ACCESS и ASSIGN. Слова обращаются (access) и назначают (assign) появившиеся ранее и эти два типа метода адресуют эти роли.

Можно было написать обычный метод возвращающий значение, как в -

METHOD nPassengers() CLASS Train

RETURN SELF:nPassengers

он может быть вызван следующим образом -

dwPass:= oTrain:nPassengers()

Это достаточно правильно, но пользователь должен знать, что надо использовать метод с ' () ', чтобы обратиться к его значению - другая сложность, программист мог сделать и без них!

Метод ACCESS разработан, чтобы решить эту проблему. Вышеупомянутый метод заменен точно таким же кодом, но заменено слово METHOD на слово ACCESS, чтобы сделать его Методом ACCESS. Итак, он станет -

ACCESS nPassengers() CLASS Train

RETURN SELF:nPassengers

Синтаксис призванный, чтобы вызвать этот тип метода не выглядит похожим на  используемый метод -

dwPass:= oTrain:nPassengers

Методы ACCESS часто используются только, чтобы выставить скрытое значение экземпляра как описано выше. Но если не предусмотрено никакого Метода ACCESS, то доступ к скрытым Ivar невозможен; доступ следовательно управляется программистом.

Методы ACCESS могут также использоваться, чтобы возвратить расчетное значение, типа этого -

ACCESS nPerCentFull() CLASS Train

RETURN SELF:nPassengers * 100 / SELF:nMaxPassengers

Он используется также, как будто Ivar, вызываем <nPerCentFull> -

nPCFull:= oTain1:nPerCentFull
Методы ACCESS для обеспечения всех Переменных Экземпляров означает, что разработчик класса может изменить, что-нибудь в классе, обеспечение внешнего интерфейса остается за ним. Если Переменные Экземпляров доступны непосредственно (ЭКСПОРТИРОВАНЫ), они могут стать необходимы для пользователя класса, чтобы изменить код и позволить длительное использование класса.

Например, возможно первоначально <nPerCentFull> был переменной экземпляра Класса Train и разработчик класса решил упростить код в классе, удалив Ivar, <nPerCentFull>, на том основании, что она была избыточна и хуже, обеспечивала возможность для противоречивости. Если Метод ACCESS для <nPerCentFull> ранее не обеспечивался, его нужно обеспечить сейчас, иначе существующий код пользователя не будет работать.

Методы ASSIGN обслуживают подобную цель. Они могут действовать как путь через барьер, обеспечивая инкапсуляцию и вызывать некоторый код для проверки правильности пути. Они могут также позволить некоторому несуществующему или виртуальному свойству быть модифицированным другими Ivars.

Метод ASSIGN написан подобно нормальному методу -

ASSIGN nMaxPassengers(nMaxPass) CLASS Train

IF nMaxPass > 4000

   Textbox{ , “Maximum No of Passengers unchanged”, ;

              “Can NEVER be more than 4000”}:Show()

ELSEIF nMaxPass < 0

   TextBox{ , “Maximum No of Passengers unchanged”, ;

              “Can NEVER be less than zero”}:Show()

ELSEIF nMaxPass < SELF:nPassengers

   TextBox{ , “Oh dear”, “Overcrowded”}:Show()

ELSE

   // The actual assignment to the IVar is made here ...

   SELF:nMaxPassengers := nMaxPass

ENDIF

RETURN SELF:nMaxPassengers

Вышеупомянутый Метод ASSIGN вызывается в следующем коде, хотя Вы не можете быть в этом уверенны -

oTrain1:nMaxPassengers := 2500

Наш Метод ASSIGN подтвердил новое значение <nMaxPassengers> и кажется является нормальным назначением. Но почему Метод ASSIGN возвращает значение? Причина в том, что она нормально должна допустить действующие назначения для переменных следующим образом -

oTrain1:nMaxPassengers := oTrain2:nMaxPassengers := 2500

В этом случае, если бы <oTrain2:nMaxPassengers> возвратил, чему-нибудь еще, <oTrain1:nMaxPassengers> дал бы неправильное значение и вероятно неправильный тип данных также. Это точно, поскольку разработчик не может знать, что используется Метод ASSIGN, что оно должно возвратить подходящее значение для использования в действующих назначениях.

Общий аспект инкапсуляции - потребность создавать свойство Только для чтения (оно не может быть полностью только для чтения, иначе ему нельзя было бы присвоить значение, но добраться до него, чтобы изменить его, могут только некоторые пользователи). Нет такого состояния для свойств «только для чтения» , но можно сделать его недоступным для пользователя, чтобы пользователь не мог изменить его значение: создайте их как СКРЫТЫЕ (HIDDEN) переменные экземпляров и обеспечьте Метод ACCESS, но никакого Метода ASSIGN (или специального с проверкой пользовательских разрешений). 

Теперь у нас есть класс, который работает и скажем, используем в настоящее время в некоторых прикладных программах, когда возникнет потребность в новом. Мы хотим более сложный Класс Train, но мы рискуем получить кодовые ошибки и уничтожение того, что уже работает. Есть лучший путь. Мы можем наследовать от нашего Класса Train и не изменять его, таким образом не воздействуя на существующее использование. Унаследованная версия будет иметь все способности существующего Класса Train и даже больше.

Наследование
Это - принцип ООП с самой высокой конфигурацией. Наследование позволяет создать новый класс со всеми свойствами и поведением существующего класса и затем позволяет его поведение расширить без изменения первоначального кода  класса. Первоначальный класс остается нетронутым и поэтому продолжает работать как прежде.

Без наследования, новый класс создавался бы путем копирования существующего класса и изменением копии. Наследование позволяет избегает дублирования (уход от ненужного увеличения в размере прикладной программы .EXE файла); наследование также избегает своего рода ошибок, которые ползают в измененной копии кода.

Инструкция класса используемая выше создала класс на пустом месте, то есть все его свойства и методы были созданы. Но мы можем создать класс, наследуя его от другого класса, если требуемый класс достаточно подобен, и можно думать как вид другой класс. Новый класс может иметь множества свойств и/или методов, которые будут функционировать точно таким же способом как родительский класс, от которого он унаследован. Мы тогда можем приступать к добавлению новых свойств и методы которые нам нужны, чтобы получить необходимое поведение для нового класса. Но мы не изменяем ничего в родительском классе; он работает, так что не ломайте его.

Унаследованные классы обычно более сложны чем те, от которых они наследуются; или по крайней мере, они должны быть. Это подразумевает некоторую предусмотрительность, необходимо изобрести более простые классы перед расширением унаследованного класса. И, конечно, класс может быть унаследован больше чем одним классом, каждый унаследованный класс имеет возможность быть расширенным для других целей. Далее, унаследованные классы могут сами быть унаследованы другими классами.

Возвращаясь к нашему Классу Train, наш пользователь теперь хочет получить статистику о пассажирах на его поездах, и наша следующая цель состоит в том, чтобы увидеть, сможем ли мы обеспечить эту информацию, сохраняя большое количество информации в каждом объекте Train и обеспечивая методы, чтобы обработать это. Для начала, мы создаем новый класс StatTrain, который унаследован от базового Класса Train -
CLASS StatTrain INHERIT Train // “Statistical Train”

HIDDEN nLegs AS DWORD // A leg is a journey section

HIDDEN nTotalDistance AS REAL4

HIDDEN nPassengerMiles as REAL4 // Related to revenue

METHOD Init(nMaxPassengers) CLASS StatTrain

SELF:nLegs := 0

SELF:nTotalDistance := 0

SELF:nPassengerMiles := 0

SUPER:Init(nMaxPassengers)

Пока, мы учитываем новые функциональные назначения Ivars для поддержку этих функциональных назначений. Но StatTrain сохраняет все функциональные возможности Train. Заметьте, что мы даже не упомянули <nPassengers>. Мы должны упомянуть <nMaxPassengers>, потому что экземпляр StatTrain требует, чтобы Train, его родитель, дали значение для <nMaxPassengers> на экземпляр. Мы сделаем егоо в пределах метода Init () метод для StatTrain, вызывая -

SUPER:Init(nMaxPassengers)

где SUPER подобен SELF, но обращается определенно к родителю текущего класса. Так два новых синтаксических понятия это - INHERIT и SUPER.
Если метод вызывается для класса, который не имеет никакого метода с таким имени, то сообщение вызывающая метод поддерживает цепочку наследования того класса до метода с таким именем, если он существуете, найден. Это относится к методу Init() подобно любой другой. В этом случае, метод Init() для класса StatTrain может быть опущен, так как он только присваивает нули значениям, которые сами являются нолем по умолчанию, и вызывает метод Init() его родителя, который вызвался бы так или иначе.

Как Вы можете видеть из того что мы охватили это, когда SELF указывается в методе, это означает то, что метод фактически вызывался для объекта унаследованного класса. Однако, объект - экземпляр как собственного класса так и своего родителя, и т.д. Так, пока <oStatTrain> возможно, присваивает значение экземпляра StatTrain, а он в свою очередь все еще экземпляр Train, так что SELF все еще применим.

Довольно неясное исключение, где SELF не обращается к объекту, которому метод был послан, где выбор из меню возвращает назад скорее собственное окно, чем унаследованное окно.

Точно так же SUPER не обязательно обращаться к одному порождению; он обращается к следующему порождению, где метод того же самого названия существует.

Наоборот, если мы вызываем метод в унаследованном классе и точно такой же метод есть в родительском классе, метод в родителе замаскирован тем же методом в ребенке. Так, если функциональные возможности метода родителя все еще требуются в ребенке, метод родителя должен вызываться явно с помощью SUPER из метода в ребенке. Если функциональные возможности метода в ребенке должны быть идентично тому самому родителю, есть конечно, нет потребности в метода в ребенке. Обратите внимание, что новый дочерний метод необходим только, если требуются различные функциональные возможности . Если новый метод включает те же самые функциональные возможности что и родительский метод, то явный запрос SUPER должен быть сделан, как мы увидим ниже в NewLeg(). Если функциональные возможности метода в родителе не соответствуют ребенку, никакой запрос SUPER не должен быть сделан.

В случае PreInit () и PostInit (), как указано в Главе 3, методы в родителе не имеют никаких функциональных возможностей, так что SUPER не требуется.

Цель этого нового класса StatTrain присвоить значение точно как прежде –

oStatTrain6:= StatTrain{2300}

Мы теперь можем добавить функциональные возможности. Идея состоит в том, что, на каждом участке пути поезда от станции к станции, он везет различное число пассажиров. Не получая слишком сложным, мы хотим быть способными найти - 
· Среднее число пассажиров в пути, 
· Среднее число PassengerMiles, который касается среднего оплаченного тарифа.

Чтобы сделать это, мы должны ввести информацию в начале каждого участка  методом разработанным для этой цели. Но так получилось, что мы уже имеем метод NewLeg() для Train; давайте адаптируем его к StatTrain -

METHOD NewLeg(nOn, nOff, nLegLength) CLASS StatTrain

IF nLegLenght <= 0

   TextBox{ , "StatTrain NewLeg Error", ;

              "Leg length must be greater than zero"}:Show()

ELSE

   IF SUPER:NewLeg(nOn, nOff)

      SELF:nLegs += 1

      SELF:nPassengerMiles += SELF:nPassengers * nLegLength

      SELF:nTotalDistance += nLegLength

   ENDIF

ENDIF

Проверка правильности <nOn> и <nOff> и модифицикация < SELF:nPassengers >, сделана в запросе SUPER, пока проверка правильности для <nLegLength> должна быть сделана здесь. Проверка правильности необходима здесь, чтобы предотвратить возможную ошибку Деления на Ноль и, конечно, <nLegLength> не может быть отрицательным. Подтвердив данные, <nLegs>, <nPassengerMiles> и <nTotalDistance> все полностью модифицированы.

Заметьте, что здесь мы использовали то же самое название метода StatTrain, что и в Train. Мы не должны иметь.
Если подходящий метод NewLeg () не был доступен в Train, мы могли бы написать весь необходимый код в StatTrain, но нам повезло (или это было за ранее запланировано?).

Теперь для статистики -

ACCESS nAvgPassengers() CLASS StatTrain

RETURN SELF:nPassengerMiles / SELF:nTotalDistance

ACCESS nAvgPassengerMiles() CLASS StatTrain

RETURN SELF:nAvgPassengers ;

             * (SELF:nTotalDistance / SELF:nLegs)

Математик может сделать запрос алгоритмов, но они удовлетворят нас здесь. Так или иначе, в любой стадии поездки объект StatTrain может cделает запрос относительно среднего числа пассажиров и средней оплаты за проезд, обе из которых рассчитывают значения раньше, чем возвращают Ivars.

Обратите внимание, что родительский класс не осведомлен о любой из его унаследованных классов. Поэтому родительский класс не может знать о любых свойствах его дочерних классов. Так, если мы вызываем метод, cодержащийся в родительском классе, мы должны помнить, что его может только включать Ivars, известный ему.

Если, в StatTrain, мы определенно хотели использовать метод NewLeg(), принадлежащий Train, он должен вызываться с помощью SUPER, чтобы избежать вызова его собственного метода с подобным именем. (Это может быть необходимо корректировать ошибку в числе пассажиров и не увеличивать число участков.)

Чуть больше, чтобы объяснить о EXPORTED и HIDDEN Ivars. Во-первых, они - не единственные параметры: есть и другие типы - INSTANCE и PROTECT. Использование правильного типа IVAR затрагивает видимость этого IVAR к другим частям вашей прикладной программы. Как объяснено выше, они должны быть защищены от неправильного употребления, но должны, конечно, выполнять свою роль.

EXPORT

Если Вы используете ЭКСПОРТИРУЕМЫЙ Ivars, то инкапсуляция утрачивается. Использование методов ACCESS и ASSIGN становятся несоответствующими, для рассчитанных значений, потому что можно было также непосредственно назначить значения. Вы могли использовать его в раннем развитии и по мере развития убирать его из вашего кода.

HIDDEN

Этот тип Ivar - не видим вне класса, в котором он определен. Так скрытый Ivar в классе Train - не видим в классе StatTrain, через ACCESS и  ASSIGN, или нормальные методы Класса Train. Используйте эту декларацию при разработке классов для библиотек, где Вы не хотите неправильное употребление вашего кода другими.

PROTECT 

Видим, только в пределах класса и в его подклассах. Видим пользователями тех классов только через методы.

INSTANCE

Очень похож на PROTECT кроме доступа и назначения Ivars того же самого класса изнутри методов того класса. В пределах класса, его Ivars доступен непосредственно, так что нет никакой потребности использовать методы ACCESS и ASSIGN, подготовленные к тем Ivars. Но если метод ссылок класса IVAR, для которого есть метод ACCESS или ASSIGN того же самого имени, должен обратиться напрямик или через инкапсуляцию методов ACCESS / ASSIGN? К IVAR, объявленному как PROTECT обратились бы непосредственно: к IVAR, объявленному как INSTANCE обратились бы через инкапсуляцию метода ACCESS / ASSIGN.

В большинстве случаев, PROTECT - декларация выбора. Однако, Вы увидите, что код, сгенерированный для Windows включает INSTANCE Ivars. Причина, по которой  сгенерированный код для окон, включает INSTANCE Ivars (которая, кажется, не имеют никакой цели) – в том, что без них, должна произойти ошибка транслятора, если Вы вызвали метод ACCESS или ASSIGN изнутри других методов класса окна без приписывания ими SELF. Но мы рекомендуем, чтобы Вы всегда использовали префикс SELF для Ivars так или иначе; это делает INSTANCE Ivars лишним.

Мы обсудили различные пути инкапсуляции Ivars и предотвратили неправомочный доступ к им. Мы можем сделать тот же самое для методов, объявляя их с инструкцией PROTECT METHOD как в -

PROTECT METHOD NewLeg(...) CLASS Train

... <code as before>

ЗАЩИЩЕННЫЕ Методы видимы только классу, в котором они определены и те, которые наследуются от него, не снаружи.

Чтобы вернуться к структурам на мгновение, класс, который не имеет никаких методов, кроме Init() и никакой вероятности быть унаследованным, может быть осуществлен как структура. Но вероятно проще заниматься классами. Структуры поддерживаются в Visual Objects для взаимодействия с Windows API и другим Си базовым кодом.

В Главе 3, мы создавали новый класс, VideoShell, который был унаследован от ShellWindow, но значение слова наследования не объяснялось в то время. Теперь, его значение должно быть понятным. Мы сделали его так, чтобы наша оболочка (VideoShell) имела свойство <oServer>, не доступным в классе ShellWindow, обеспеченном в языке.

Полиморфизм
Это последнее из четырех принципов ООП и вероятно самое простое в понимании. Оно просто заявляет, что различные классы могут иметь методы, совместно использующие то же самое имя. Мы делали это выше с NewLeg () и как Train и как StatTrain классифицирует методы с этим название, которые являются полностью независимыми за исключением StatTrain:NewLeg() вызывая SUPER:NewLeg(). Это означает, что, при создании нового класса, Вы не должны волноваться о том, как Вы называете его методы. Однако, это не совсем так просто как есть, некоторое преимущество будет получено называя тщательно методы в некоторых случаях.
Очевидно удобно что общие действия типа дисплея, сохранять, и печать может использоваться как названия метода многих классов.

Это могло быть удобно, если создать ряд классов, загруженных в массив и все с методом Print(). Тогда единственный цикл массива может печатать каждого в свою очередь -

METHOD Print() CLASS PRINTCOLLECTION

LOCAL i as DWORD

FOR i := 1 UPTO ALEN(SELF:aObjects)

    SELF:aObjects[i]:Print()

NEXT

Снова в Главе 3 мы использовали полиморфизм, в методе Close() применялся и к ShellWindow и к объектам DB Server. И VideoShell и Video1S, унаследованный от классов, которые уже имели методы Close(), так что мы могли использовать их без того, чтобы bметь необходимость определять что случаются внутри этих методов. Однако, мы были должны использовать слово SUPER, чтобы гарантировать, что, когда VideoShell был закрыт, его вызвал методом Close() его родитель ShellWindow, потому что мы создали метод в VideoShell с тем же самым названием как метод в его родителе. (Обратите внимание, что мы не могли использовать, любой имя кроме Close() в этом случае, потому что этот метод всегда вызывается системой всякий раз, когда пользователь закрывает прикладную программу, и мы должны убедиться, что выявили в эту цепочку событий.)

Обзор
На этом завершаем наше обсуждение о четырех принципах Объектно-Ориентированного Программирования. Немного тяжело, но полное понимание всего этого - основа звукового объектно-ориентированного программирования.

Мы видели, как Абстракция позволяет нам создавать типы данных, которые отражают объекты в реальном мире.

Обсуждение по Инкапсуляции показало нам, как скрыть детали выполнения где нужно.

Новые классы могут быть созданы от существующих, используя Наследование, новый класс, имеют все свойства и поведение оригинала.

Наконец, Полиморфизм позволяет нам имена метода повторно использовать в различных классах.

Это вероятно стоит повторить, если Вы не поняли все это, Вы постепенно начнете  понимать это с использованием.

